
Change 1 

	Function name
	f_IMS_MO_Call_504_ServerTimeout

	Reason for change
	After the UE receives 504 Server Time out, the UE sends ACK and TCP Close and immediately TCP SYNC in order to establish the next TCP connection for port 5060 and the expected SIP Request.  TCP socket is listening, therefore TCP Accept is sent to TTCN. At this time TTCN is still having the old TCP connection on port 5060 and discards the TCP Accept. Therefore the expected SIP Request is not received in TTCN and the test case fails. So, there is a race condition between TTCN configuration and UE behaviour on TCP socket handling. In order to avoid the race condition it is proposed to first release the TCP handling configuration  in IP/ IMS PTC and then abort the procedure in IPCAN PTC


	Summary of change
	In order to initiate first the Relase of the old TCP connection, changed the relative ordering of calls to f_IMS_GlobalRelease and f _IMS_CC_AbortCall.


	TTCN module
	IMS_CC_TestcaseImplementation_12.ttcn

	MCC160 Comment
	


Before change
  function f_IMS_MO_Call_504_ServerTimeout(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4(-, p_AudioCodecList);

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_Ims_3gpp_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_CC_AbortCall();    // to abort call establishment procedure

    f_IMS_GlobalRelease (oldestProtectedContextOnly); // @sic R5s170468 - additional change sic@
    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f_IMS_TestBody_Set(false);

  }
After Change 
  function f_IMS_MO_Call_504_ServerTimeout(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC

  {

    var template (present) MessageHeader v_MessageHeader_Ack;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;

    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;  /* @sic R5w140112: IMS_InviteRequestWithSdp_Type instead of IMS_DATA_REQ */

    var INVITE_Request v_InviteRequest;

    var template (present) SipUrl v_RequestSipUrl := f_SIP_BuildSipUri_RX(px_IMS_CalleeUri);

    f_IMS_TestBody_Set(true);

    // Step 1 - Make the UE attempt an IMS speech call

    // @siclog "Step 1" siclog@

    f_IMS_CC_StartCall(IPCAN_MO_SpeechCall);

    // Step 2. Receive INVITE Request

    // @siclog "Step 2" siclog@

    v_InviteRequestWithSdp := f_IMS_MOCallSetup_AnnexC21_Step2_A4(-, p_AudioCodecList);

    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);

    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Step 3. SS sends 504 Server Time-out.

    // @siclog "Step 3" siclog@

    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine504,

                                                                   f_IMS_CommonResponse_504_MessageHeaderTX(v_InviteRequest.msgHeader),

                                                                   cs_MessageBody_XML(cs_Ims_3gpp_Restoration))));

    // Step 4. Receive ACK

    // @siclog "Step 4" siclog@

    v_MessageHeader_Ack := f_IMS_AckRequest_MessageHeaderRX(v_InviteRequest, c_statusLine504);

    f_IMS_ReceiveRequest(car_IMS_Ack_Request(cr_ACK_Request(v_RequestSipUrl, v_MessageHeader_Ack)));

    f_IMS_GlobalRelease (oldestProtectedContextOnly); // @sic R5s170468 - additional change sic@

    f_IMS_CC_AbortCall();    // to abort call establishment procedure
    // Steps 5 to 12. UE performs initial registration Annex C.2

    // @siclog "Step 5 - 12" siclog@

    f_IMS_Registration();

    f _IMS_TestBody_Set(false);

  } 

